
1. GRUNDLAGEN

Wenn sich zu einer gegebenen Funktion f keine Stammfunktion angeben läßt, können wir im allge-
meinen das Integral

                                              
a

b

∫ f (x) dx

nicht exakt berechnen. Es existieren aber verschiedene Verfahren, mit deren Hilfe sich der Wert des
Integrales beliebig gut approximieren läßt. Jede Obersumme oder jede Untersumme ist beispielswei-
se eine solche Approximation. Rechteck-, Trapez-, und Simpsonregel basieren auf derselben Idee -
die Annäherung des Integrals durch den Flächeninhalt von Rechtecken oder Trapezen - liefern aber
mit Hilfe einfacher geometrischer Überlegungen bessere Ergebnisse.

1.1. RECHTECKREGEL

Die Fläche unter der Kurve wird durch ein Rechteck mit den Seitenlängen (b-a) und ƒ(
a b+

2
) ange-

nähert.

             a                       a b+
2

                       b

 
a

b

∫ f (x) dx ≈  R ( f (x), a, b)  =  f ((a+b)/2)* (b-a)

Durch Unterteilung des Intervalls [a;b] in n gleichgroße Intervalle der Länge h = 
b a

n
−

, Anwendung

der Rechteckregel auf jedes Intervall [a + i*h; a + (i+1)*h]  h=0,1,...,n-1 und anschließender Sum-
mation wird die Näherung noch verbessert. Es ergibt sich folgende Formel:

RECHTECK  (ƒ(x),a, b, n) =  h * 
i

n

=

−

∑
0

1

ƒ (a + (i + 0.5)*h)

1.2. TRAPEZREGEL

Die Fläche unter der Kurve wird angenähert durch ein Trapez mit der Höhe (b-a) und den Seitenlän-
gen ƒ(a) und ƒ(b)

               a                                  b

a

b

∫ f (x) dx ≈  T ( f (x), a, b)  =  
f a f b( ) ( )+

2
 * (b-a)

Unterteilung des Intervalles [a;b] in n Teilintervalle der Länge h = 
b a

n
−

, wiederholte Anwendung der

Trapezregel und anschließende Summation führt zu

TRAPEZ (ƒ(x),a,b,n) = 
h
2

 * [ƒ(a) + ƒ(b) +
1

1n−

∑ f (a + i*h) ]

1.3. SIMPSON-REGEL

Bei Rechteck- und Trapezregel wird der Funktionsverlauf durch Geraden angenähert. Dabei werden
ein bzw. zwei Stützwerte benutzt, um den Verlauf der Geraden festzulegen. Die Simpsonregel ver-
wendet 3 Stützwerte und nähert den Funktionsverlauf durch eine Parabel an.
Die Fläche unter dieser Parabel ist dann der Näherungswert für das gesuchte Integral.

            a                  
2

ba +                b

Andererseits läßt sich die Simpsonregel auch als kombinierte Anwendung von Rechteck- und Tra-
pezregel erklären: Auf das Intervall [a;b] wird die Rechteckregel für n=1 und die Trapezregel für n=2
angewendet. Die so erhaltenen Werte werden dementsprechend im Verhältnis 1 : 2 gewichtet. Die
Zahl der zu berechnenden Funktionswerte bleibt somit gegenüber der Trapezregel gleich, die Ge-
nauigkeit der Näherung wird aber im allgemeinen verbessert. Es ergibt sich

a

b

∫ f (x) dx ≈  S (ƒ((x),a,b) = 1
3

 [ RECHTECK (ƒ(x),a,b,1)  + 2 * TRAPEZ (ƒ(x),a,b,2)]

                                         = 
b a−

6
 [ƒ (a) + 4*ƒ (0.5*(a+b)) + ƒ(b)]

(Bemerkung: Dieser Spezialfall der Simpsonregel mit 3 Stützwerten ist als Keplersche Faßregel be-
kannt)

Unterteilung des Intervalls [a;b] in n Teilintervalle der Länge h = 
b a

n
−

  wiederholte Anwendung der

Simpson-Regel führt und anschließende Summation führt zu:

SIMPSON (ƒ(x),a,b,n) = 
h
3

  ( S0 + 2*S1 + 4*S2 )         mit S0  := ƒ (a)  +  ƒ (b)

                                                                                           S1 := 
0< <
∑
i n

i gerade

ƒ (a + i*h)

                                                                                           S2 := 
0< <
∑
i n

i ungerade

ƒ (a + i*h)

2. DIE NÄHERUNGSVERFAHREN AUF DEM TI-92

2.1. DEFINITION DER NÄHERUNGSVERFAHREN AUF DEM TI-92

Die Formeln für die Näherungsverfahren lassen sich mit Hilfe von Define  (Menü F4) im Home-
Editor als Funktionen definieren. Dabei werden nur die Intervallgrenzen a,b und die Unterteilungszahl
n als Variablen behandelt. Die betreffende Funktion muß unabhängig unter dem Namen f(x) definiert
werden.

• Definition der Funktion rechteck (a,b,n) im
Home-Editor:
 define rechteck (a,b,n) = (b-a)/n*S(f(a+(i+0.5)*
 (b-a)/n),i,0,n-1)

 
• Definition der Funktion trapez (a,b,n) im

Home-Editor:
 define trapez (a,b,n) = (b-a)/(2*n)*(f(a)+f(b)+2*
 S(f(a+i*(b-a)/n),i,1,n-1))  

 
 Im Programm-Editor lassen sich die Formeln auch schrittweise und somit übersichtlicher definieren,
was vor allem für die Definition der Simpson-Formel nützlich ist.
 
• Start des Programm-Editors durch APPS -
7-3

• Wahl der Option Function im Feld Type
• Eingabe des Funktionsnamens rechteck im

Feld Variable
• Verlassen des New-Fensters durch Enter

• Definition der Funktion rechteck (a,b,n) im
Programm-Editor

 

     (Der Befehl Local befindet sich im Menü
       F4)

 
 
 Auf die gleiche Weise definiert man die anderen Formeln im Programm-Editor.
 
 trapez (a,b,n)
 Func
  Local h,t0,t1,i
  (b-a)/n→h
  f(a)+f(b) →t0
  S(f(a+i*h),i,1,n-1) →t1
  h/2*(t0+2*t1)
 EndFunc

 simpson (a,b,n)
 Func
  Local h,i,s0,s1,s2
  (b-a)/n→h
  f(a)+f(b)→s0
  0→s1
  For i,2,n-1,2
    s1+f(a+i*h)→s1
  EndFor
  0→s2
  For i,1,n-1,2
    s2+f(a+i*h)→s2
  EndFor
  h/3*(s0+2*s1+4*s2)
 EndFunc

 

 2.2. VERGLEICH DER NÄHERUNGSVERFAHREN
 
 Nachdem wir die Näherungsformeln im Home-Editor oder im Programm-Editor definiert haben, sind
wir jetzt in der Lage, vom Home-Editor aus beliebige Funktionen in beliebigen Integrationsgrenzen
und mit beliebiger Genauigkeit näherungsweise zu integrieren. Dazu definieren wir zunächst die zu
integrierende Funktion unter dem Namen f(x) und rufen anschließend das gewünschte Näherungs-
verfahren mit den entsprechenden Parametern auf.
 
• Definition der Funktion f(x)=x2

• Berechnung des Integrals f
−
∫

1

1

(x) dx

       - exakt durch die ∫ - Funktion

          (Eingabe von ∫ (f(x),x,-1,1) )

       - näherungsweise mit der Funktion rechteck
       -                          mit der Funktion trapez
       -                          mit der Funktion simpson

 

 
 Um die Qualität der einzelnen Näherungsverfahren besser vergleichen zu können, bietet es sich an,
eine Tabelle für verschiedene Werte von n anzulegen. Dies kann über den Tabellen-Editor gesche-
hen. Zunächst definieren wir eine zu integrierende Funktion f(x) im Home-Editor (im Beispiel ist f(x)
= x4 + 2x3 - 2x2 + x + 2).
 Da die Variable n nur ganzzahlige Werte annimmt, wechseln wir dann vom Function-Mode in
den Sequence-Mode, in dem mit Folgen statt mit Funktionen gearbeitet wird.

• Aufrufe des Mode-Fensters durch MODE
• Wahl der Option SEQUENCE im Feld Graph
• Verlassen des Mode-Fensters durch ENTER

 
 

 Anschließend rufen wir den y=Editor auf.
 

• Aufruf des y=Editors durch APPS-2
• Eingabe der Ausdrücke
       u1(n) = rechteck (0,4,n)
       u2(n) = trapez (0,4,n)
       u3(n) = simpson (0,4,n)

 
 

 Nun können wir uns im Tabellen-Editor eine Tabelle erstellen lassen, in der die im y=Editor eingege-
benen Ausdrücke für verschiedene Werte von n errechnet werden. Da die Simpson-Regel nur bei
einer geraden Unterteilungszahl sinvoll anwendbar ist, lassen wir n nur gerade Werte annehmen.
 

• Aufruf des Tabellen-Editors durch
APPS - 5

• Aufruf des Fensters TABLE-SETUP
durch F2

• Eingabe des Startwertes 2 im Feld
tblStart

• Eingabe der Schrittweite 2 im Feld ∆tbl
• Verlassen des Fensters Table-Setup

durch Enter
 

 
 
 

 Wir erhalten nebenstehende Tabelle.
 

 In Spalte 1 befinden sich die verschiedenen Un-
terteilungszahlen; die Spalten 2 - 4 enthalten die
entsprechenden Näherungswerte von Rechteck-,
Trapez-, und Simpsonregel.

 
 

 Die so erhaltene Tabelle zeigt, daß für alle Verfahren die Werte sich mit wachsender Unterteilungs-
zahl dem exakten Integralwert nähern. Natürlich lassen sich die Parameter, die in die Tabelle einflie-
ßen, beliebig modifizieren:



NUMERISCHE INTEGRATION MIT RECHTECK-, TRAPEZ-, UND SIMPSONREGEL

                                                                                                                                                                                                          
Thomas Schmidt, Stromberg

7

 

• Die zu integrierende Funktion kann im Home-Editor durch define f(x)=... umdefiniert werden.
Ein erneuter Aufruf des Tabellen-Editors liefert dann automatisch die Näherungswerte für diese
Funktion

 
• Durch Ändern der Schrittweite (Fenster Table-Setup, Feld ∆tbl) von 2 auf 1 zeigt die

Tabelle auch Näherungswerte für ungerade Unterteilungszahlen an (dabei zeigt sich auch, daß die
Simpson-Regel für ungerade Unterteilungszahlen keine brauchbaren Näherungswerte liefert)

 
• Die Integrationsgrenzen können durch Änderung der y=Ausdrücke verändert (z.B. u2(n)=trapez

(1,3,n)) werden.
 
 
 Weiterhin können wir ausnutzen, daß wir die Näherungswerte in einer Folge definiert haben und uns

diese graphisch darstellen lassen. Dazu müssen wir das Grafik-Fenster erst mit geeigneten  Gren-
zen für die x-Achse und die y-Achse versehen.

 
• Aufruf des Window-Editors durch
APPS-3

• Eingabe der Werte
        nmin = 0;  nmax = 12
        plotstrt =2 ; plotstep = 2
        xmin = 0; xmax = 0
        xscl = 2
        ymin = 290; ymax = 350
        yscl = 5

 
 
 Bei einem anschließenden Aufruf des Graph-
Fensters (APPS-4) ergibt sich nebenstehendes
Bild, aus dem man ersehen kann, wie die mit den
drei Näherungsverfahren errechneten Werte sich
mit wachsender Unterteilungszahl untereinander
annähern und einem Grenzwert zustreben.

 
 
 
 2.3. GRAPHISCHE VERANSCHAULICHUNG DER NÄHERUNGSVERFAHREN
 
 Mit Hilfe eines Programmes, das die Befehle graph und line  verwendet, können wir  graphisch ver-
anschaulichen, wie das Integral mit Hilfe der Rechteckregel angenähert wird.
 Vor Beginn sollten alle eventuell im y=Editor vorhandenen Ausdrücke durch
 
        APPS-2-F1-8-ENTER
 
 gelöscht werden, damit sie beim Aufruf von graph nicht automatisch gezeichnet werden.
 Außerdem sollte im Mode-Fenster für das Feld Graph die Option FUNCTION gewählt sein
(MODE - → - 1 - ENTER).
 

 2.3.1. GRAPHISCHE VERANSCHAULICHUNG DER RECHTECKREGEL

 Die Eingabe des Programmes erfolgt über den Programm-Editor.
 

• Aufruf des Programm-Editors durch APPS -
7-3

• Wahl der Option Program im Feld Type
• Eingabe des Programmnamens drawrcht im

Feld Variable
• Eingabe des Programmtextes im Programm-

Editor (vollständiger Programmtext s. unten
rechts)

 
 

 

 

 drawrcht (a,b,n)
 Prgm
  Local h,i
 @ FESTLEGUNG DES X-ACHSEN-BEREICHES
  a-(b-a)/10→xmin
  b+(b-a)/10→xmax
 @ ZEICHNEN DER FUNKTION
  ClrDraw
  Graph f(x)
  (b-a)/n→h
 @  ZEICHNEN DER RECHTECKE
  For i,0,n-1
   Line a+i*h,0,a+i*h,f(a+(i+0.5)*h)
   Line a+(i+1)*h,0,a+(i+1)*h,f(a+(i+0.5)*h)
   Line a+i*h,f(a+(i+0.5)*h),a+(i+1)*h,f(a+(i+0.5)*h)
  EndFor
 EndPrgm

 

 Vor dem Aufruf des Programmes empfiehlt es sich, über den Window-Editor einen geeigneten y-
Achsen-Bereich (der natürlich von f(x) und den Integrationsgrenzen abhängig ist) für das Grafik-
Fenster festzulegen. Der x-Achsen-Bereich wird vom Programm automatisch passend zu den Inte-
grationsgrenzen festgelegt.
 

• Aufruf des Window-Editors durch
APPS - 3

• Festlegung der unteren Grenze des y-Achsen-
Bereiches unter ymin

• Festlegung der oberen Grenze des y-Achsen-
Bereiches unter ymax

• Gegebenenfalls Festlegung der Achsenskalie-
rung unter xscl und yscl  

 
 Der Aufruf des Programmes erfolgt über den Home-Editor. Wiederum muß vorher die zu integrie-
rende Funktion unter dem Namen f(x) definiert worden sein.

• Aufruf des Home-Editors durch
APPS - 1

• Definition der Funktion f(x) durch
        define f(x) = x^2 + 1
• Aufruf des Programmes drawrcht mit Angabe

der Integrationsgrenzen und der Unterteilungs-
zahl

 
 
 Wir erhalten nebenstehendes Bild.
 Wenn wir das Programm erneut aufrufen und
dabei eine größere Unterteilungszahl eingeben,
wird deutlich, wie die Näherung verbessert wird.
 
 drawrcht (-1,1,20)        (unten links)
 drawrcht (-1,1,50)   (unten rechts)

 
 

  
 

 

 .3.2. GRAPHISCHE VERANSCHAULICHUNG DER TRAPEZREGEL
 

 Nach demselben Prinzip können wir ein Programm zur Veranschaulichung der Trapezregel schreiben:
 
• Aufruf des Programm-Editors durch APPS -
7-3

• Wahl der Option Program im Feld Type
• Eingabe des Programmnamens drawtrap im

Feld Variable
• Eingabe des nebenstehenden Programmtextes

im Programm-Editor

 drawtrap (a,b,n)
 Prgm
  Local h,i
 @ FESTLEGUNG DES X-ACHSEN-BEREICHES
  a-(b-a)/10→xmin
  b+(b-a)/10→xmax
 @ ZEICHNEN DER FUNKTION
  ClrDraw
  Graph f(x)
  (b-a)/n→h
 @ ZEICHNEN DER TRAPEZE
  For i,0,n-1
   Line a+i*h,0,a+i*h,f(a+i*h)
   Line a+(i+1)*h,0,a+(i+1)*h,f(a+(i+1)*h)
   Line a+i*h,f(a+i*h),a+(i+1)*h,f(a+(i+1)*h)
  EndFor
 EndPrgm

 Mit f(x) = x2 + 1
        ymin = -0.5
 und ymax = 2.5
 ergibt sich bei Aufruf von
 drawtrap (-1,1,3)
 nebenstehendes Bild.

 
 

 

 2.3.3. GRAPHISCHE VERANSCHAULICHUNG DER SIMPSONREGEL
 

 Eine graphische Veranschaulichung der Simpsonregel gestaltet sich etwas schwieriger. Wir können
einerseits die zugrundeliegende Anwendung der Rechteck- und Trapezregel verdeutlichen. Dazu
bietet es sich an, den Split-Screen-Modus zu verwenden, der es erlaubt zwei getrennte Grafiken auf
dem Bildschirm zu erstellen.
 Folgendes Programm veranschaulicht das Zustandekommen der Simpsonregel als Kombination aus
Rechteck- und Trapezregel im Split-Screen.
 
 
• Aufruf des Programm-Editors durch APPS -
7-3

• Wahl der Option Program im Feld Type
• Eingabe des Programmnamens drawsimp im

Feld Variable
• Eingabe des nebenstehenden Programmtextes

im Programm-Editor

 drawsimp(a,b,n)
 Prgm
  Local oben,unten
 @ AUFTEILEN DES BILDSCHIRMES
  setMode("Split Screen","LEFT-RIGHT")
  setMode("Number of Graphs","2")
  setMode("Graph 2","FUNCTION")
  ymin→unten
  ymax→oben
 @ RECHTECKREGEL IN FENSTER 1
  drawrcht(a,b,n/2)
 @ TRAPEZREGEL IN FENSTER 2
  switch(2)
  unten→ymin
  oben→ymax
  drawtrap(a,b,n)
 EndPrgm

 
 
 Mit f(x) = x3 - 2x2  + x + 3
        ymin = -0.5
 und ymax = 5
 ergibt sich bei Aufruf von drawsimp(-0.6,1.5,4) folgendes Bild. Dabei wird in der linken
Bildschirmhälfte die Anwendung der Rechteckregel für n=2 verdeutlicht, in der rechten Bildschirm-
hälfte die Anwendung der Trapezregel für die doppelte Unterteilungszahl n=4.

 Nach Programmende bleibt der Rechner im
Split-Screen-Modus. Um zum normalen Modus
zurückzukehren, geht man wie folgt vor:
• Aufruf der 2.Seite des Mode-Fensters durch
MODE - F2

• Wahl der Option FULL im Feld Split
Screen

 
 
 Andererseits läßt sich die Simpsonregel auch unabhängig von den beiden anderen Näherungsverfah-
ren veranschaulichen, indem man die Näherungsparabel errechnet und zeichnet. Die gesuchte Parabel
hat die Form
 

                                                 parabel (x) = k1x2 + k2x +k3

 und enthält die Punkte (a, f(a)), (p, f(p)) und (b, f(b)), wobei  p = 
a b+

2
. Zur Bestimmung der (ein-

deutig bestimmten) Koeffizienten k1,k2 und k3 müssen wir also das Gleichungssystem
 

                                                 k1 a2   +   k2 a   +   k3   =   f(a)
                                                 k1 p2   +   k2 p   +   k3   =   f(p)
                                                 k1 b2   +   k2 b   +   k3   =   f(b)           lösen.
 

 Dies kann mit Hilfe des Befehles simult geschehen. Folgendes Programm errechnet erst die Nähe-
rungsparabel und zeichnet dann die zu integrierende Funktion und die Näherungsparabel in ein Ko-
ordinatensystem. Für letztere wird dabei der Befehl shade  verwendet, um die Fläche unter der Pa-
rabel herauszuheben.
 

• Aufruf des Programm-Editors durch
APPS - 7- 3

• Wahl der Option Program im Feld Type
• Eingabe des Programmnamens drawsmp2 im

Feld Variable
• Eingabe des nebenstehenden Programmtextes

im Programm-Editor

 drawsmp2(a,b)
 Prgm
 @ FESTLEGUNG DES X-ACHSEN-BEREICHES
   a-(b-a)/10→xmin
   b+(b-a)/10→xmax
 @ ERRECHNEN DER NÄHERUNGSPARABEL
 si-
mult([[a^2,a,1][((a+b)/2)^2,(a+b)/2,1][b^2,b,1]],[[f(a)][f((a
+b)/2)][f(b)]])→k

 Define parabel(x)= ∑(k[3-i,1]*x^i,i,0,2)
 @ ZEICHNEN DER FUNKTIONEN
  ClrDraw
  Graph f(x)
  Shade 0,parabel(x),a,b
 EndPrgm
 
 

 

 Das Programm behandelt nur die Intervallgrenzen a und b als Variablen. n ist fest auf 2 gesetzt, die zu
integrierende Funktion f(x) muß wiederum vorher definiert worden sein. Außerdem empfiehlt es sich
auch diesmal vor Aufruf des Programmes einen geeigneten y-Achsen-Bereich im Window-Editor
festzulegen.
 

 Mit f(x) = x3 - x2 + x + 6,
        ymin = -5
 und ymax = 30
 ergibt sich bei Aufruf von
 drawsmp2 (-1,3)
 nebenstehendes Bild.

 
 

 2.4. GRENZWERTUNTERSUCHUNGEN ZU DEN NÄHERUNGSVERFAHREN
 

 Im allgemeinen nähern sich die Werte von Rechteck-, Trapez- und Simpsonregel mit wachsender
Unterteilungszahl n immer weiter dem exakten Integralwert an. Wir können dies an Einzelbeispielen
überprüfen. Neben der bereits beschriebenen Möglichkeit, eine Tabelle für verschiedene Werte von
n anzulegen, bietet sich auch die Verwendung der Funktion limit an, mit deren Hilfe sich Grenzwerte
berechnen lassen.
 Dabei ist jedoch Vorsicht geboten, da der TI-92 nicht in jedem Falle das erwartete Ergebniss liefert,
sondern in vielen Fällen den eingegebenen Ausdruck einfach unbearbeitet wieder ausgibt, wenn ihm
eine Grenzwertberechnung nicht möglich ist.
 Dies ist der Fall, wenn mit schrittweise definierten Funktionen gearbeitet wird. Wir geben deshalb die
Funktionen rechteck(a,b,n) und trapez(a,b,n) in einem Schritt über den Home-Editor ein (siehe
2.1.1.). Die Simpson-Regel kann aufgrund ihrer Komplexität überhaupt nicht mit dem Befehl limit
bearbeitet werden (Grenzwertuntersuchungen zur Simpson-Regel siehe Aufgabe 3.1.)
 Weiterhin können die Berechnungen mit limit auch dann scheitern, wenn die zu integrierende Funkti-
on für den TI nicht geeignet ist, um einen Grenzwert zu bestimmen. Dies ist zum Beispiel schon bei so
einfachen Funktionen wie sin(x) der Fall.
 Trotzdem lassen sich beispielsweise für Polynome die erwarteten Ergebnisse bestätigen:

• Definition von f(x) = x5 + 2x3 - 2
 define f(x)=x^5+2*x^3-2

• Berechnung des exakten Integralwertes für die
Integrationsgrenzen 1 und 3
 ∫ (f(x),x,1,3)

• Berechnung der Grenzwerte von rechteck
und trapez

      limit (rechteck (1,3,n),n,∞)
       limit (trapez (1,3,n),n, ∞)

 

 

 

 3. AUFGABEN UND ERWEITERUNGEN
 

 3.1. Definieren Sie Funktionen obersum(a,b,n) und untersum(a,b,n), die zu einer gegebenen (mo-
noton wachsenden) Funktion die Obersumme bzw. Untersumme bei einer Unterteilung des Intervalls
[a;b] in n gleichgroße Teilintervalle berechnen.
 
 Lösungsvorschlag: a) Definition der Funktionen im Home-Editor durch
                                    define obersum (a,b,n) = ∑ (f (a + i*(b-a)/n)*(b-a)/n, i, 1, n)
                                    define untersum (a,b,n) = ∑ (f (a + i*(b-a)/n)*(b-a)/n, i, 0, n-1)        oder
 
                                b) Schrittweise Definition der Funktionen im Funktionen Editor
 
 obersum(a,b,n)
 Func
  Local h
  (b-a)/n→h
  ∑ (h*f(a+i*h),i,1,n)
 EndFunc
 

 untersum (a,b,n)
 Func
  Local h
  (b-a)/n→h
  ∑ (h*f(a+i*h),i,0,n-1)
 EndFunc

 
 3.2. Schreiben Sie ein Programm zur graphischen Veranschaulichung von Ober- und Untersumme.
 
 Lösungsvorschlag: Eingabe folgender Programme im Programm-Editor. Da die Programme sich nur in 2 Zeilen
                                von drawrcht unterscheiden, bietet es sich an, dessen Programmtext umzuschreiben, anstatt
                                den kompletten Text einfach neu einzugeben.
 
 drawuntr (a,b,n)
 Prgm
  Local h,i
 @ FESTLEGUNG DES X-ACHSEN-BEREICHES
  a-(b-a)/10→xmin
  b+(b-a)/10→xmax
 @ ZEICHNEN DER FUNKTION
  ClrDraw
  Graph f(x)
  (b-a)/n→h
 @ ZEICHNEN DER RECHTECKE
  For i,0,n-1
   Line a+i*h,0,a+i*h,f(a+i*h)
   Line a+(i+1)*h,0,a+(i+1)*h,f(a+i*h)
   Line a+i*h,f(a+i*h),a+(i+1)*h,f(a+i*h)
  EndFor
 EndPrgm

 drawober (a,b,n)
 Prgm
  Local h,i
 @ FESTLEGUNG DES X-ACHSEN-BEREICHES
  a-(b-a)/10→xmin
  b+(b-a)/10→xmax
 @ ZEICHNEN DER FUNKTION
  ClrDraw
  Graph f(x)
  (b-a)/n→h
 @ ZEICHNEN DER RECHTECKE
  For i,0,n-1
   Line a+i*h,0,a+i*h,f(a+(i+1)*h)
   Line a+(i+1)*h,0,a+(i+1)*h,f(a+(i+1)*h)
   Line a+i*h,f(a+(i+1)*h),a+(i+1)*h,f(a+(i+1)*h)
  EndFor
 EndPrgm

 Mit f(x)=ln x,
         ymin=-0.5 und
         ymax=2.5
 ergibt sich bei Aufruf von drawuntr (1,8,10) bzw. drawober (1,8,10) folgendes Bild:
 

  
 

 

 3.3. Überprüfen sie anhand einer Tabelle und einer oder mehrerer Beispielfunktionen folgende Be-
hauptungen:
 

 3.3.1. Die mit der Simpson-Formel errechneten Werte nähern sich mit wachsender Unterteilungszahl
dem exakten Integralwert an.
 3.3.2. Die Simpsonregel ergibt sich durch
             SIMPSON (f((x),a,b,2n) = 1

3
 [ RECHTECK (f(x),a,b,n)  + 2 * TRAPEZ (f(x),a,b,2n)]

           aus Rechteck- und Trapezregel.
 
 Lösungsvorschlag: 1. Definition einer Beispielfunktion ƒ (x) = sin (x) im Home-Editor
                               2. Aufruf des Mode-Fensters und Wahl der Option SEQUENCE im Feld Graph (s. Kap. 2.2.)
                                3.  Aufruf des y=Editor
                                4. Eingabe der Ausdrücke
                                     u1(n) = simpson (0,π,n)

                                     u2(n) = abs (simpson (0,π,n) - ∫ (f(x),x,0, π))

                                     u3(n) = 1
3

 (rechteck (0, π,n/2) + 2*trapez (0, π,n))

                                5.  Aufruf des Tabellen-Editors
                                6. Eingabe des Startwertes tblStart = 4 und der Schrittweite ∆tbl = 4
                                    im Setup-Fenster
                                7. Verlassen des Setup-Fensters durch ENTER
               (8. Ggf. Zurücksetzen des graph-Modes auf FUNCTION)

 3.3.3.  Für eine monoton wachsende Funktion ist der mit der Trapezformel errechnete Wert gleich
dem Mittelwert von Ober- und Untersumme.
 
 Lösungsvorschlag: 1. Definition einer Beispielfunktion ƒ (x) = ln (x) im Home-Editor
                               2. Aufruf des Mode-Fensters und Wahl der Option SEQUENCE im Feld Graph (s. Kap. 2.2.)
                                3. Aufruf des y=Editors
                                4. Eingabe der Ausdrücke
                                     u1(n) = trapez ( 1, 10, n)
                                     u2(n) = 0.5*(obersum (1,10,n) + untersum (1,10,n))
                                5.  Aufruf des Tabellen-Editors
                                6. Eingabe des Startwertes tblStart = 1 und der Schrittweite ∆tbl = 1
                                    im Setup-Fenster
                                7. Verlassen des Setup-Fensters durch ENTER
                               (8. Ggf. Zurücksetzen des graph-Modes auf FUNCTION)
 
                                    (Bemerkung: die Funktionen obersum und untersum müssen vorher definiert worden sein
                                     siehe Aufgabe 1)
 

  
 
 3.4. Definieren Sie eine Funktion simpson2 (a,b,n), die die Simpson-Näherung durch Integration der
Näherungsparabeln errechnet.
 
 Lösungsvorschlag: Definition der Funktion im Programm-Editor
 
• Start des Programm-Editors durch
APPS - 7-3

• Wahl der Option Function im Feld
Type

• Eingabe des Funktionsnamens simpson2 im
Feld Variable

• Definition der Funktion simpson (a,b,n) im
Programm-Editor durch Eingabe des neben-
stehenden Programmtextes

 
 

 simpson2(a,b,n)
 Func
  Local h,i,p1,p2,p3,wert,koeff,p(x)
  (b-a)/n→h
  0→wert
  For i,0,n-2,2
   a+i*h→p1
   a+(i+1)*h→p2
   a+(i+2)*h→p3
 si-
mult([[p1^2,p1,1][p2^2,p2,1][p3^2,p3,1]],[[f(p1)][f(p2)][f(
p3)]]) →koeff
  Define p(x)=∑(koeff[3-j,1]*x^j,j,0,2)

   wert+ ∫ (p(x),x,p1,p3)→wert
  EndFor
  wert
 EndFunc

 Stichproben zeigen, daß die von simpson (a,b,n)
und simpson2 (a,b,n) errechneten Werte gleich
sind. Natürlich kann über den Tabellen-Editor
auch ein umfangreicherer Test vorgenommen
werden.

 
 

 3.5. Zeigen Sie für den Spezialfall f(x)=x2, daß

 untersumme (f(x),a,b,n) n→∞ →
1
3

b3  -  
1
3

a3

 obersumme (f(x),a,b,n) n→∞ →
1
3

b3  -  
1
3

a3

          (Hauptsatz der Integralrechnung)
 
 Lösungsvorschlag: 1. Definition der Funktionen untersum (a,b,n) und obersum (a,b,n) im Home-Editor (!)
                                    (s. Aufgabe 3.1.)
                                2. Definition der Funktion f(x)=x2

                                3. Berechnung des Grenzwertes für die Untersumme durch
                                    limit(untersum(a,b,n),n,∞)
                                4. Ausmultiplizieren des Ergebnisausdruckes durch expand(ans(1))
                                5. Berechnung des Grenzwertes für die Obersumme durch
                                    limit(obersum(a,b,n),n,∞)
                                6. Ausmultiplizieren des Ergebnisausdruckes durch expand(ans(1))

 GRUNDEINSTELLUNGEN ZUR ARBEIT MIT DEN ‘WERKZEUGEN’
 

 Damit die im folgenden dargestellten Programme, Funktionen und Berechnungen wie beschrieben
und fehlerlos arbeiten, sollten vor Beginn einige Grundeinstellungen am TI vorgenommen werden:
 

• Um Speicherplatz freizugeben, sollte entweder ein Memory-Reset vorgenommen werden, oder
zumindest nicht benötigte Variablen freigegeben werden. In jedem Falle sollten die Variablen
a,b,c,...,z unbelegt sein.

 
 DURCHFÜHREN EINES MEMORY-RESET
 (alle selbstdefinierten Objekte werden gelöscht
!!!)
 
• Aufruf des MEMORY-Fensters durch 2nd-6
• Aufruf des Menüs RESET (F1) und Wahl von 1:

All oder 2: Memory

 
  

 LÖSCHEN EINZELNER VARIABLEN
 
• Aufruf des VAR-LINK-Fensters durch 2nd-

Minustaste
• Auswahl der zu löschenden Variablen durch F4
• Aufruf von Delete im Menü Manage (F1)
• Bestätigen durch ENTER

 
 

 

 

 FREIGEBEN DER VARIABLEN a,b,...,z
 
• Aufruf von Clear a-z (F6) im Home-Editor
• Bestätigen durch ENTER

 
 
• Im MODE-Fenster (Aufruf durch MODE) sollten folgende Modi gesetzt sein:

• Sofern nicht ausdrücklich anders erwähnt, wird im FUNCTION-Modus gearbeitet. Bei Um-
schalten in den SEQUENCE-Modus sollte, sofern nicht anders erwähnt im Menü AXES (F7)
des y=Editors die Option TIME gewählt sein.

 
• Eventuell im y=Editor vorhandene Ausdrücke sollten gelöscht oder deselektiert sein.

VERZEICHNIS DER FUNKTIONEN UND PROGRAMME

NAME ARBEITSWEISE BEMERKUNGEN
DRAWOBER (a,b,n) EINGABEN:

   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
ARBEITSWEISE:
   Zeichnet die Funktion ƒ(x) im
   Graph-Fenster und veranschaulicht
   graphisch die Obersumme für die
   Unterteilungszahl n

Die betreffende Funktion muß
vor Aufruf unter dem Namen
ƒ(x) definiert worden sein. Der
y-Achsen-Bereich muß im
Window-Editor geeignet fest-
gelegt werden

DRAWRCHT (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
ARBEITSWEISE:
   Zeichnet die Funktion ƒ(x) im
   Graph-Fenster und veranschaulicht
   graphisch die Rechteckregel für die
   Unterteilungszahl n.

Die betreffende Funktion muß
vor Aufruf unter dem Namen
ƒ(x) definiert worden sein. Der
y-Achsen-Bereich muß im
Window-Editor geeignet fest-
gelegt werden

DRAWSIMP (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
ARBEITSWEISE:
   Ruft den Split-Screen auf,
   zeichnet die Funktion ƒ(x) in beiden
   Graph-Fenstern und veranschaulicht
   graphisch das Zustandekommen der
   Simpsonregel als Kombination aus
   Rechteck- und Trapezregel für die
   Unterteilungszahl n.

Die betreffende Funktion muß
vor Aufruf unter dem Namen
ƒ(x) definiert worden sein. Der
y-Achsen-Bereich muß im
Window-Editor geeignet fest-
gelegt werden
Nach Beendigung des Pro-
grammes bleibt der Rechner im
Split-Screen-Modus.

DRAWSMP2 (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
ARBEITSWEISE:
   Zeichnet die Funktion ƒ(x) im
   Graph-Fenster und veranschaulicht
   graphisch die der Simpsonregel zu-
   grundeliegende Integration der
   Näherungsparabel

Die betreffende Funktion muß
vor Aufruf unter dem Namen
ƒ(x) definiert worden sein. Der
y-Achsen-Bereich muß im
Window-Editor geeignet fest-
gelegt werden

DRAWTRAP (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
ARBEITSWEISE:
   Zeichnet die Funktion ƒ(x) im
   Graph-Fenster und veranschaulicht
   graphisch die Trapezregel für die
   Unterteilungszahl n.

Die betreffende Funktion muß
vor Aufruf unter dem Namen
ƒ(x) definiert worden sein. Der
y-Achsen-Bereich muß im
Window-Editor geeignet fest-
gelegt werden

NAME ARBEITSWEISE BEMERKUNGEN
DRAWUNTR (a,b,n) EINGABEN:

   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
ARBEITSWEISE:
   Zeichnet die Funktion ƒ(x) im
   Graph-Fenster und veranschaulicht
   graphisch die Untersumme für die
   Unterteilungszahl n

Die betreffende Funktion muß
vor Aufruf unter dem Namen
ƒ(x) definiert worden sein. Der
y-Achsen-Bereich muß im
Window-Editor geeignet fest-
gelegt werden

OBERSUM (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
AUSGABE:
   Obersumme der Funktion ƒ(x) über
   dem Intervall [a;b] bei Unterteilung
   in n Teilintervalle

Die betreffende Funktion muß
vor Aufruf unter dem Namen
ƒ(x) definiert worden sein.

RECHTECK (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
AUSGABE:
  Näherungswert für das Integral der
  Funktion ƒ(x) in den Grenzen a und b
  bei Approximation mit Hilfe der
  Rechteckregel und Unterteilung des
  Intervalls in n Teilintervalle

Die zu integrierende Funktion
muß vor Aufruf unter dem
Namen ƒ(x) definiert worden
sein.

SIMPSON (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
AUSGABE:
  Näherungswert für das Integral der
  Funktion ƒ(x) in den Grenzen a und b
  bei Approximation mit Hilfe der
  Simpsonregel und Unterteilung des
  Intervalls in n Teilintervalle

Die zu integrierende Funktion
muß vor Aufruf unter dem
Namen ƒ(x) definiert worden
sein.

SIMPSON2 (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
AUSGABE:
  Näherungswert für das Integral der
  Funktion ƒ(x) in den Grenzen a und b
  bei Approximation mit Hilfe der
  Simpsonregel (berechnet durch
  Integration der Näherungsparabeln)
  und Unterteilung des Intervalls in n
  Teilintervalle

Die betreffende Funktion muß
vor Aufruf unter dem Namen
ƒ(x) definiert worden sein.
Die Berechnung ist bei gro-
ßem n sehr zeitaufwendig!

TRAPEZ (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls

Die zu integrierende Funktion
muß vor Aufruf unter dem
Namen ƒ(x) definiert worden

NAME ARBEITSWEISE BEMERKUNGEN
   n : Unterteilungszahl
AUSGABE:
  Näherungswert für das Integral der
  Funktion ƒ(x) in den Grenzen a und b
  bei Approximation mit Hilfe der
  Trapezregel und Unterteilung des
  Intervalls in n Teilintervalle

sein.

UNTERSUM (a,b,n) EINGABEN:
   a : linker Rand des Intervalls
   b : rechter Rand des Intervalls
   n : Unterteilungszahl
AUSGABE:
   Untersumme der Funktion ƒ(x) über
   dem Intervall [a;b] bei Unterteilung
   in n Teilintervalle

Die betreffende Funktion muß
vor Aufruf unter dem Namen
ƒ(x) definiert worden sein.


